
Abstract

In the context of formal cryptographic protocol verification, logging-
independent message anonymity is the property for a given message to
remain anonymous despite the attacker’s capability of mapping mes-
sages of that sort to agents based on some intrinsic feature of such
messages, rather than by logging the messages exchanged by legiti-
mate agents as with logging-dependent message anonymity.

This paper illustrates how logging-independent message anonymity
can be formalized according to the relational method for formal pro-
tocol verification by considering a real-world protocol, namely the Re-
stricted Identification one by the BSI. This sample model is used to
verify that the pseudonymous identifiers output by user identification
tokens remain anonymous under the expected conditions.

1 Logging-independent message anonymity and Re-
stricted Identification

theory Definitions
imports Main

begin

1.1 Introduction

Logging-dependent message anonymity is the property for a message ex-
changed or otherwise used in a cryptographic protocol to remain anony-
mous although the attacker can log the messages generated or accepted by
legitimate agents, and map any two observable messages contained in any
such message to the same agent. An approach to modeling and verifying
this security property according to the relational method for formal protocol
verification has been described in [7], along with the method itself. This ap-
proach makes use of two further type constructors for messages, IDInfo and
Log, as well as of two functions, crypts and key-sets. Particularly, IDInfo is
used to model message anonymity, while the remaining constants are used
to formalize the property for a message to be observable by the spy within
some logged message.
Logging-independent message anonymity rather is the property for a given
message to remain anonymous in spite of the attacker’s capability of map-
ping messages of that sort to agents without resorting to message logging,
namely by means of some intrinsic feature of such messages. From the above
observation, it follows that IDInfo is the sole anonymity-related constant re-
quired if the only kind of anonymity of interest for a given protocol is the
logging-independent one, whereas Log, crypts, and key-sets are unnecessary
and can be left out of the model. It is also possible to include both kinds of
anonymity in the model, in which case some protocol rule will enable the spy
to map messages of some sort to agents only if they are observable within
logged messages, while some other protocol rule will enable the spy to do so
independently of this condition.
This paper illustrates how logging-independent message anonymity can be
formalized according to the relational method by considering a real-world
protocol, namely the Restricted Identification one by the BSI [1] [2], whose
very purpose is to allow for the exchange of messages endowed with this
security property.

1.2 Case study: the Restricted Identification protocol

The Restricted Identification protocol enables user identification tokens (e.g.
electronic documents) to generate and output unambiguous pseudonymous
identifiers, distinct for any given group of terminals of arbitrary granularity,

referred to as a sector, and usable to identify the tokens across different
sessions taking place within the same sector. For example, such identifiers
allow for the creation of sector-specific revocation lists, at the same time
preserving the anonymity of the holder of any token included in such a list.
This protocol is based on a Public Key Infrastructure (PKI) comprising the
token issuer, which owns a revocation key pair (SKRev, PKRev) and gen-
erates a token key pair (SKTok, PKTok) for each token, and sectors, each
one endowed with its own sector key pair (SKSec, PKSec), where PKSec =
[SKSec]PKRev. This PKI may use either an integer finite field or an elliptic
curve group, as long as the selected domain parameters are cryptographi-
cally secure. In a real-world PKI, each sector has actually two distinct key
pairs, which enables the tokens to generate as many different pseudonymous
identifiers per sector, but this detail is irrelevant to the anonymity of such
identifiers and can then be omitted from the model.
According to [1] [2], the Restricted Identification protocol may only be ex-
ecuted using the session keys established via Chip Authentication version
2/3, after performing Terminal Authentication version 2 with a terminal
certificate containing both sector public keys’ hashes (including domain pa-
rameters), whose authenticity is ensured by the certificate’s signature. Af-
ter requesting to start the protocol, which again is irrelevant and can be
left out of the model, the terminal sends either of its sector public keys
PKSec (including domain parameters) to the token, which in turn veri-
fies that PKSec’s hash matches the one contained in the certificate and
replies with its pseudonymous identifier H([SKTok]PKSec), where H is a
hash function. If necessary (for instance, to insert it into a sector-specific
revocation list), this identifier can be recomputed in the external world as
H([SKSec]([SKRev]PKTok)), with the concurrence of both the token issuer
and the entity responsible for the involved sector.
As a matter of fact, since the only purpose of the protocol model to be devel-
oped is to verify the logging-independent anonymity of token pseudonymous
identifiers, without any confidentiality or authenticity concern, it is sufficient
to model a simpler protocol in which the terminal and the token exchange
their messages in plain, without using any session key. Moreover, since
both Chip and Terminal Authentication are out of scope, the Restricted
Identification protocol will be modeled as a stand-alone one. Consequently,
although both of them are exchanged during Terminal Authentication in
the real-world protocol, PKSec’s signature will be assumed to be exchanged
with PKSec in the model, while the related verification key will be assumed
to be known by the token a priori. The hash function used to sign PKSec

may differ from the one used to compute token pseudonymous identifiers,
but once more, this is just an omissible functional detail.
A further simplification, admissible for the same reason, is to let the token
use domain parameters known a priori rather than the input ones, whose

presence in the input message can then be left out. Indeed, this prevents
the spy from snatching SKTok by making an element of a smaller group
pass for an authentic sector public key, which could be done by signing it
with a compromised signature generation key. For example, if the PKI used
a group of 128-bit order, SKTok could be disclosed by first searching the
private key SK ′

Tok associated with a fake identifier ranging in a group of 64-
bit order n, and then detecting SKTok as the unique private key associated
with a given genuine identifier among all those differing from SK ′

Tok by a
multiple of n. So, two searches within as many spaces of 264 elements, which
is a computationally feasible task nowadays, would suffice to find SKTok.
However, such small group attacks can safely be ruled out as long as the
initial state s0 comprises an arbitrary set of compromised token private keys,
given that verifying the conditions under which these keys remain secret is
out of scope.
As a result of all the simplifications described above, the protocol that is
going to be modeled is as follows.

1. Terminal → Token: {PKSec, {H(PKSec)}SKSign
}

The terminal sends the token a message consisting of its sector public
key and a precomputed signature of this key.

2. Token → Terminal: H([SKTok]PKSec)

The token verifies that the hash of the received public key matches the
signed one, and then replies with the pseudonymous identifier resulting
from this key.

Unless it is compromised by means other than attacking the protocol, the
anonymity of a given token pseudonymous identifier H([SKTok]PKSec) is ex-
pected to be vulnerable if and only if either SKTok is anonymity-compromised,
or SKSec is compromised and there is another compromised sector pri-
vate key SK ′

Sec such that H([SKTok]PK ′
Sec) is anonymity-compromised.

In fact, the spy can detect the use of SKTok in H([SKTok]PKSec) in the
former case, whereas in the latter one he can map H([SKTok]PKSec) to
the same token as H([SKTok]PK ′

Sec) by recognizing that [SKTok]PKSec =
[SKSec × (SK ′

Sec)
−1]([SKTok]PK ′

Sec).
The purpose of the following formal development is precisely to formally
prove the correctness of this expectation. In more detail, the only if con-
ditional implied by the previous statement will be proven as an anonymity
property in the next section, while the if conditional will be proven in the
form of two possibility properties in the subsequent one. Since both relevant

attack options leverage the intrinsic features of token pseudonymous iden-
tifiers, namely the private keys used to generate them, logging-independent
message anonymity has to be considered rather than logging-dependent one.
For further information about the formal definitions and proofs contained
in this paper, see Isabelle documentation, particularly [6], [5], [3], and [4].

1.3 Agents, messages, protocol rules

Agents consist of an infinite population of tokens and sectors, identified
through natural numbers, plus the spy. Actually, the model can safely ignore
terminals altogether and assume that tokens are presented to sectors as a
whole, since all the terminal-side messages used in the protocol refer to
sectors rather than to individual terminals. The only possible exceptions are
signature key pairs. In fact, although the entity signing terminal certificates
is the same for every terminal in a given sector (in [1] and [2], it is named
Document Verifier), nothing prevents that entity to use distinct signature
generation keys for different terminals (for example, if their certificates are
issued at different times). Nonetheless, the granularity of signature key pairs
is irrelevant to the anonymity of token pseudonymous identifiers according
to the previous considerations, so these key pairs can be associated with
sectors as well.
As opposed to what happens in [7], there is no correlation here between any
two agents Token n and Sector n marked with the same numeric identifier
n. In fact, tokens and sectors are independent of each other, and any token
may be presented to whatever sector.

type-synonym agent-id = nat

datatype agent =
Token agent-id |
Sector agent-id |
Spy

As regards the key pairs for key agreement, private keys are identified by
natural numbers, whereas public keys by sets of natural numbers. The
implied interpretation is that PubK S stands for public key [k]G, where
G is the group generator and k is the modular product of all the private
keys referred to by the numeric identifiers in S, each one occurring as a
factor exactly once. Using multisets of natural numbers instead of sets
would have allowed private keys to be used as factors even more than once,
but this option can be left out as the PKI does not provide for any public
key computed in this way. The need for this ad hoc message format, like
those used to represent session keys and Chip Authentication Data in [7],

confirms that reuse of the spy’s capabilities’ model in the inductive method
is hindered by the likely need for ad hoc message formats in case of protocols
using nontrivial public key cryptography [7].
Besides key agreement keys, messages comprise signature generation/verifi-
cation keys (identified by the numeric identifiers of the respective sectors),
hash values, cryptograms, and compound messages built via message con-
catenation. Furthermore, message anonymity is modeled by means of con-
structor IDInfo [7]. Since the anonymity of terminal-side messages is of no
concern, the interpretation of message 〈n, X〉 is "message X is mapped to
Token n", namely n is always interpreted as a token’s numeric identifier
rather than a sector’s one.

type-synonym key-id = nat

datatype agr-key =
PriK key-id |
PubK key-id set

datatype enc-key =
SigK agent-id |
VerK agent-id

datatype msg =
AgrKey agr-key |
EncKey enc-key |
Hash msg |
Crypt enc-key msg |
MPair msg msg |
IDInfo agent-id msg

syntax
-MPair :: [ ′a, args] ⇒ ′a ∗ ′b (‹(2{|-,/ -|})›)
-IDInfo :: [agent-id, msg] ⇒ msg (‹(2 〈-,/ -〉)›)

syntax-consts
-MPair ⇀↽ MPair and
-IDInfo ⇀↽ IDInfo

translations
{|X , Y , Z |} ⇀↽ {|X , {|Y , Z |}|}
{|X , Y |} ⇀↽ CONST MPair X Y
〈n, X〉 ⇀↽ CONST IDInfo n X

abbreviation SigKey :: agent-id ⇒ msg where
SigKey ≡ EncKey ◦ SigK

abbreviation VerKey :: agent-id ⇒ msg where
VerKey ≡ EncKey ◦ VerK

abbreviation PriKey :: key-id ⇒ msg where
PriKey ≡ AgrKey ◦ PriK

abbreviation PubKey :: key-id set ⇒ msg where
PubKey ≡ AgrKey ◦ PubK

primrec InvK :: enc-key ⇒ enc-key where
InvK (SigK n) = VerK n |
InvK (VerK n) = SigK n

abbreviation InvKey :: enc-key ⇒ msg where
InvKey ≡ EncKey ◦ InvK

inductive-set parts :: msg set ⇒ msg set
for H :: msg set where

parts-used [intro]:
X ∈ H =⇒ X ∈ parts H |

parts-crypt [intro]:
Crypt K X ∈ parts H =⇒ X ∈ parts H |

parts-fst [intro]:
{|X , Y |} ∈ parts H =⇒ X ∈ parts H |

parts-snd [intro]:
{|X , Y |} ∈ parts H =⇒ Y ∈ parts H

definition parts-msg :: msg ⇒ msg set where
parts-msg X ≡ parts {X}

Constant Rev-PriK is the numeric identifier of the revocation private key,
while functions Sec-PriK and Tok-PriK map the numeric identifiers of sec-
tors and tokens to those of the respective sector/token private keys. It is
assumed that these functions are injective, as well as that their ranges do
not contain Rev-PriK and are disjoint, and such axioms are proven to be
consistent by showing that there exist three constants satisfying all of them.
On the whole, these axioms just model the assumption that private keys are
generated by cryptographically secure means throughout the PKI, so that
the probability for any given private key to occur more than once within the
PKI is negligible.

consts Rev-PriK :: key-id

consts Sec-PriK :: agent-id ⇒ key-id

consts Tok-PriK :: agent-id ⇒ key-id

specification (Rev-PriK Sec-PriK Tok-PriK )
sec-prik-inj: inj Sec-PriK
tok-prik-inj: inj Tok-PriK
sec-prik-rev: Rev-PriK /∈ range Sec-PriK
tok-prik-rev: Rev-PriK /∈ range Tok-PriK
sec-prik-tok-prik: range Sec-PriK ∩ range Tok-PriK = {}

by (rule exI [of - 0 ], rule exI [of - λn. n + n + 1 ],
rule exI [of - λn. n + n + 2 ], auto simp: inj-on-def , arith)

abbreviation Gen-PubKey :: msg where
Gen-PubKey ≡ PubKey {}

abbreviation Rev-PriKey :: msg where
Rev-PriKey ≡ PriKey Rev-PriK

abbreviation Rev-PubKey :: msg where
Rev-PubKey ≡ PubKey {Rev-PriK}

abbreviation Tok-PriKey :: agent-id ⇒ msg where
Tok-PriKey n ≡ PriKey (Tok-PriK n)

abbreviation Tok-PubKey :: agent-id ⇒ msg where
Tok-PubKey n ≡ PubKey {Tok-PriK n}

abbreviation Sec-PriKey :: agent-id ⇒ msg where
Sec-PriKey n ≡ PriKey (Sec-PriK n)

abbreviation Sec-PubKey :: agent-id ⇒ msg where
Sec-PubKey n ≡ PubKey {Sec-PriK n, Rev-PriK}

abbreviation Sign :: agent-id ⇒ msg ⇒ msg where
Sign n X ≡ Crypt (SigK n) (Hash X)

abbreviation ID :: agent-id ⇒ msg ⇒ msg where
ID n X ≡ case X of AgrKey (PubK S) ⇒ PubKey (insert (Tok-PriK n) S)

The spy’s starting knowledge, as defined by the initial state s0, consists of
the following messages.

• All the public keys used in the PKI (including the group generator).

• All token pseudonymous identifiers (in both hashed and non-hashed
formats).

• Compromised private keys used in the PKI (excluding the revocation
one, assumed to be secret).

• Mappings of all token public keys, compromised token private keys,
and anonymity-compromised token pseudonymous identifiers (in both
hashed and non-hashed formats) to the respective tokens.

consts bad-sigk :: agent-id set

consts bad-sec-prik :: agent-id set

consts bad-tok-prik :: agent-id set

consts bad-id :: (agent-id × agent-id) set

type-synonym event = agent × msg

type-synonym state = event set

abbreviation used :: state ⇒ msg set where
used s ≡ Range s

abbreviation spied :: state ⇒ msg set where
spied s ≡ s ‘‘ {Spy}

abbreviation s0 :: state where
s0 ≡ {Spy} × ({Gen-PubKey, Rev-PubKey} ∪

SigKey ‘ bad-sigk ∪ Sec-PriKey ‘ bad-sec-prik ∪ Tok-PriKey ‘ bad-tok-prik ∪
range VerKey ∪ range Sec-PubKey ∪ range Tok-PubKey ∪
range (λ(n, m). ID n (Sec-PubKey m)) ∪
range (λ(n, m). Hash (ID n (Sec-PubKey m))) ∪
range (λn. 〈n, Tok-PubKey n〉) ∪
{〈n, Tok-PriKey n〉 | n. n ∈ bad-tok-prik} ∪
{〈n, ID n (Sec-PubKey m)〉 | n m. (n, m) ∈ bad-id} ∪
{〈n, Hash (ID n (Sec-PubKey m))〉 | n m. (n, m) ∈ bad-id})

Protocol rules are defined here below. Particularly, for any public key [SK1×
...×SKn]G known to the spy, they enable him to generate public key [SK1×
... × SKn × SKn+1]G for any additional, compromised private key SKn+1,
as well as public key [SK1 × ... × SKi−1 × SKi+1 × ... × SKn]G for any
compromised private key SKi, where 1 ≤ i ≤ n (which is equivalent to
multiplying the original public key by (SKi)

−1). The spy can also map the
resulting public keys to the same token, if identified, as the original public
key, in the latter case as long as the related token private key still occurs as a
factor in the resulting modular product. Furthermore, the spy can associate
a token with any known public key whose modular product of private keys

contains the corresponding token private key as a factor, provided that this
key is compromised.

abbreviation rel-sector :: (state × state) set where
rel-sector ≡ {(s, s ′) | s s ′ m.

s ′ = s ∪ {Sector m, Spy} × {{|Sec-PubKey m, Sign m (Sec-PubKey m)|}}}

abbreviation rel-token :: (state × state) set where
rel-token ≡ {(s, s ′) | s s ′ m n S .

s ′ = s ∪ {Token n, Spy} × {Hash (ID n (PubKey S))} ∧
{|PubKey S , Sign m (PubKey S)|} ∈ used s}

abbreviation rel-pubk-less :: (state × state) set where
rel-pubk-less ≡ {(s, s ′) | s s ′ A S .

s ′ = insert (Spy, PubKey (S − {A})) s ∧
{PriKey A, PubKey S} ⊆ spied s}

abbreviation rel-pubk-more :: (state × state) set where
rel-pubk-more ≡ {(s, s ′) | s s ′ A S .

s ′ = insert (Spy, PubKey (insert A S)) s ∧
{PriKey A, PubKey S} ⊆ spied s}

abbreviation rel-hash :: (state × state) set where
rel-hash ≡ {(s, s ′) | s s ′ X .

s ′ = insert (Spy, Hash X) s ∧
X ∈ spied s}

abbreviation rel-dec :: (state × state) set where
rel-dec ≡ {(s, s ′) | s s ′ K X .

s ′ = insert (Spy, X) s ∧
{Crypt K X , InvKey K} ⊆ spied s}

abbreviation rel-enc :: (state × state) set where
rel-enc ≡ {(s, s ′) | s s ′ K X .

s ′ = insert (Spy, Crypt K X) s ∧
{X , EncKey K} ⊆ spied s}

abbreviation rel-sep :: (state × state) set where
rel-sep ≡ {(s, s ′) | s s ′ X Y .

s ′ = s ∪ {Spy} × {X , Y } ∧
{|X , Y |} ∈ spied s}

abbreviation rel-con :: (state × state) set where
rel-con ≡ {(s, s ′) | s s ′ X Y .

s ′ = insert (Spy, {|X , Y |}) s ∧
{X , Y } ⊆ spied s}

abbreviation rel-id-pubk-less :: (state × state) set where
rel-id-pubk-less ≡ {(s, s ′) | s s ′ n A S .

s ′ = insert (Spy, 〈n, PubKey (S − {A})〉) s ∧
{PriKey A, PubKey (S − {A}), 〈n, PubKey S〉} ⊆ spied s ∧
Tok-PriK n ∈ S − {A}}

abbreviation rel-id-pubk-more :: (state × state) set where
rel-id-pubk-more ≡ {(s, s ′) | s s ′ n A S .

s ′ = insert (Spy, 〈n, PubKey (insert A S)〉) s ∧
{PriKey A, PubKey (insert A S), 〈n, PubKey S〉} ⊆ spied s}

abbreviation rel-id-pubk-prik :: (state × state) set where
rel-id-pubk-prik ≡ {(s, s ′) | s s ′ n S .

s ′ = insert (Spy, 〈n, PubKey S〉) s ∧
{Tok-PriKey n, PubKey S} ⊆ spied s ∧
Tok-PriK n ∈ S}

abbreviation rel-id-hash :: (state × state) set where
rel-id-hash ≡ {(s, s ′) | s s ′ n X .

s ′ = s ∪ {Spy} × {〈n, X〉, 〈n, Hash X〉} ∧
{X , Hash X} ⊆ spied s ∧
(〈n, X〉 ∈ spied s ∨ 〈n, Hash X〉 ∈ spied s)}

definition rel :: (state × state) set where
rel ≡ rel-sector ∪ rel-token ∪ rel-pubk-less ∪ rel-pubk-more ∪

rel-hash ∪ rel-dec ∪ rel-enc ∪ rel-sep ∪ rel-con ∪
rel-id-pubk-less ∪ rel-id-pubk-more ∪ rel-id-pubk-prik ∪ rel-id-hash

abbreviation in-rel :: state ⇒ state ⇒ bool (infix ‹`› 60 ) where
s ` s ′ ≡ (s, s ′) ∈ rel

abbreviation in-rel-rtrancl :: state ⇒ state ⇒ bool (infix ‹|=› 60 ) where
s |= s ′ ≡ (s, s ′) ∈ rel∗

end

2 Anonymity of token pseudonymous identifiers
theory Anonymity

imports Definitions
begin

This section contains a proof of anonymity property id-anonymous, which
states that a token pseudonymous identifier remains anonymous if its anonymity
is not compromised by means other than attacking the protocol and neither
attack option described in section 1.2 is viable. As shown here below, this

property can be proven by applying rules rtrancl-induct and rtrancl-start in
a suitable combination [7].

proposition rtrancl-start [rule-format]:
(x, y) ∈ r∗ =⇒ P y −→ ¬ P x −→

(∃ u v. (x, u) ∈ r∗ ∧ (u, v) ∈ r ∧ (v, y) ∈ r∗ ∧ ¬ P u ∧ P v)
(is - =⇒ - −→ - −→ (∃ u v. ?P2 x y u v))

proof (erule rtrancl-induct, simp, (rule impI )+)
fix y z
assume

A: (x, y) ∈ r∗ and
B: (y, z) ∈ r and
C : P z

assume P y −→ ¬ P x −→(∃ u v. ?P2 x y u v) and ¬ P x
hence D: P y −→ (∃ u v. ?P2 x y u v) by simp
show ∃ u v. ?P2 x z u v
proof (cases P y)

case True
with D obtain u v where ?P2 x y u v by blast
moreover from this and B have (v, z) ∈ r∗ by auto
ultimately show ?thesis by blast

next
case False
with A and B and C have ?P2 x z y z by simp
thus ?thesis by blast

qed
qed

proposition state-subset:
s |= s ′ =⇒ s ⊆ s ′

by (erule rtrancl-induct, auto simp add: rel-def image-def )

proposition spied-subset:
s |= s ′ =⇒ spied s ⊆ spied s ′

by (rule Image-mono, erule state-subset, simp)

proposition parts-init:
parts (used s0) = used s0

by (rule equalityI , rule-tac [!] subsetI , erule-tac [2 ] parts-used,
erule parts.induct, auto)

proposition parts-idem [simp]:
parts (parts H ) = parts H

by (rule equalityI , rule subsetI , erule parts.induct, auto)

proposition parts-mono:
H ⊆ H ′ =⇒ parts H ⊆ parts H ′

by (rule subsetI , erule parts.induct, auto)

lemma parts-union-1 :
parts (H ∪ H ′) ⊆ parts H ∪ parts H ′

by (rule subsetI , erule parts.induct, auto)

lemma parts-union-2 :
parts H ∪ parts H ′ ⊆ parts (H ∪ H ′)

by (rule subsetI , erule UnE , erule-tac [!] parts.induct, auto)

proposition parts-union [simp]:
parts (H ∪ H ′) = parts H ∪ parts H ′

by (rule equalityI , rule parts-union-1 , rule parts-union-2 )

proposition parts-insert:
parts (insert X H ) = parts-msg X ∪ parts H

by (simp only: insert-def parts-union, subst parts-msg-def , simp)

proposition parts-msg-mono:
X ∈ H =⇒ parts-msg X ⊆ parts H

by (subgoal-tac {X} ⊆ H , subst parts-msg-def , erule parts-mono, simp)

proposition parts-msg-agrkey [simp]:
parts-msg (AgrKey K ) = {AgrKey K}

by (subst parts-msg-def , rule equalityI , rule subsetI , erule parts.induct, auto)

proposition parts-msg-hash [simp]:
parts-msg (Hash X) = {Hash X}

by (subst parts-msg-def , rule equalityI , rule subsetI , erule parts.induct, auto)

lemma parts-crypt-1 :
parts {Crypt K X} ⊆ insert (Crypt K X) (parts {X})

by (rule subsetI , erule parts.induct, auto)

lemma parts-crypt-2 :
insert (Crypt K X) (parts {X}) ⊆ parts {Crypt K X}

by (rule subsetI , simp, erule disjE , blast, erule parts.induct, auto)

proposition parts-msg-crypt [simp]:
parts-msg (Crypt K X) = insert (Crypt K X) (parts-msg X)

by (simp add: parts-msg-def , rule equalityI , rule parts-crypt-1 , rule parts-crypt-2 )

lemma parts-mpair-1 :
parts {{|X , Y |}} ⊆ insert {|X , Y |} (parts {X} ∪ parts {Y })

by (rule subsetI , erule parts.induct, auto)

lemma parts-mpair-2 :
insert {|X , Y |} (parts {X} ∪ parts {Y }) ⊆ parts {{|X , Y |}}

by (rule subsetI , simp, erule disjE , blast, erule disjE , erule-tac [!] parts.induct,

auto)

proposition parts-msg-mpair [simp]:
parts-msg {|X , Y |} = insert {|X , Y |} (parts-msg X ∪ parts-msg Y )

by (simp add: parts-msg-def , rule equalityI , rule parts-mpair-1 , rule parts-mpair-2 )

proposition parts-msg-idinfo [simp]:
parts-msg 〈n, X〉 = {〈n, X〉}

by (subst parts-msg-def , rule equalityI , rule subsetI , erule parts.induct, auto)

proposition parts-msg-parts:
[[(A, X) ∈ s; Y ∈ parts-msg X ]] =⇒ Y ∈ parts (used s)

by (subgoal-tac X ∈ parts (used s), drule parts-msg-mono [of X ], auto)

proposition prikey-spied:
[[s0 |= s; PriKey K ∈ parts (used s)]] =⇒ PriKey K ∈ spied s

by (induction rule: rtrancl-induct, subst (asm) parts-init,
auto simp: rel-def parts-insert dest!: parts-msg-parts)

proposition prikey-crypt [simplified]:
[[(Spy, Crypt K (PriKey K ′)) ∈ s; s0 |= s]] =⇒ PriKey K ′ ∈ spied s

by (erule prikey-spied, blast)

proposition prikey-mpair-fst [simplified]:
[[(Spy, {|PriKey K , Y |}) ∈ s; s0 |= s]] =⇒ PriKey K ∈ spied s

by (erule prikey-spied, blast)

proposition prikey-mpair-snd [simplified]:
[[(Spy, {|Y , PriKey K |}) ∈ s; s0 |= s]] =⇒ PriKey K ∈ spied s

by (erule prikey-spied, blast)

proposition rev-prikey-secret:
s0 |= s =⇒ Rev-PriKey /∈ spied s

by (induction rule: rtrancl-induct, insert sec-prik-rev tok-prik-rev,
auto simp: rel-def dest: prikey-crypt prikey-mpair-fst prikey-mpair-snd)

proposition sec-prikey-secret:
[[s0 |= s; n /∈ bad-sec-prik]] =⇒ Sec-PriKey n /∈ spied s

by (induction rule: rtrancl-induct, insert sec-prik-inj sec-prik-tok-prik, auto simp:
rel-def inj-on-def image-def dest: prikey-crypt prikey-mpair-fst prikey-mpair-snd)

proposition tok-prikey-secret:
[[s0 |= s; n /∈ bad-tok-prik]] =⇒ Tok-PriKey n /∈ spied s

by (induction rule: rtrancl-induct, insert tok-prik-inj sec-prik-tok-prik, auto simp:
rel-def inj-on-def image-def dest: prikey-crypt prikey-mpair-fst prikey-mpair-snd)

proposition idinfo-spied:

[[s0 |= s; 〈n, X〉 ∈ parts (used s)]] =⇒ 〈n, X〉 ∈ spied s
by (induction rule: rtrancl-induct, subst (asm) parts-init,
auto simp: rel-def parts-insert dest!: parts-msg-parts)

proposition idinfo-crypt:
[[(Spy, Crypt K 〈n, X〉) ∈ s; s0 |= s]] =⇒ 〈n, X〉 ∈ spied s

by (erule idinfo-spied, blast)

proposition idinfo-mpair-fst:
[[(Spy, {|〈n, X〉, Y |}) ∈ s; s0 |= s]] =⇒ 〈n, X〉 ∈ spied s

by (erule idinfo-spied, blast)

proposition idinfo-mpair-snd:
[[(Spy, {|Y , 〈n, X〉|}) ∈ s; s0 |= s]] =⇒ 〈n, X〉 ∈ spied s

by (erule idinfo-spied, blast)

proposition idinfo-hash-hash [rotated]:
[[s0 |= s; (Spy, 〈n, Hash (Hash X)〉) ∈ s]] =⇒ 〈n, Hash X〉 ∈ spied s

by (induction arbitrary: X rule: rtrancl-induct, auto simp: rel-def
dest: idinfo-crypt idinfo-mpair-fst idinfo-mpair-snd)

proposition sec-prik-eq:
{Tok-PriK n, Sec-PriK m, Rev-PriK} =

{Tok-PriK n, Sec-PriK m ′, Rev-PriK} =⇒ m ′ = m
by (erule equalityE , drule subsetD [where c = Sec-PriK m], simp, insert
sec-prik-inj sec-prik-rev sec-prik-tok-prik, auto simp: inj-on-def image-def )

proposition id-identified:
assumes

A: s0 |= s and
B: (n, m) /∈ bad-id and
C : n /∈ bad-tok-prik and
D: 〈n, Hash (ID n (Sec-PubKey m))〉 ∈ spied s

shows m ∈ bad-sec-prik ∧
(∃m ′. m ′ 6= m ∧ m ′ ∈ bad-sec-prik ∧ (n, m ′) ∈ bad-id)

proof −
let ?P1 = λs. 〈n, Hash (ID n (Sec-PubKey m))〉 ∈ spied s
let ?P2 = λs. ∃S . 〈n, PubKey S〉 ∈ spied s ∧ Sec-PriK m ∈ S
let ?P3 = λS s. 〈n, Hash (PubKey S)〉 ∈ spied s
let ?P4 = λS s. 〈n, PubKey S〉 ∈ spied s ∧ Rev-PriK ∈ S ∧
(∀m. Sec-PriK m ∈ S −→ (n, m) /∈ bad-id)

have E : ∀m. Sec-PriK m 6= Rev-PriK
by (rule allI , rule notI , subgoal-tac Rev-PriK ∈ range Sec-PriK ,
simp add: sec-prik-rev, rule range-eqI , rule sym)

have ∃ u v. s0 |= u ∧ u ` v ∧ v |= s ∧ ¬ ?P1 u ∧ ?P1 v
using A and B and D by (rule-tac rtrancl-start, auto dest: sec-prik-eq)

then obtain u1 v1 where F : s0 |= u1 ∧ u1 ` v1 ∧ ¬ ?P1 u1 ∧ ?P1 v1

by blast

moreover from this have G: 〈n, ID n (Sec-PubKey m)〉 ∈ spied u1

by (auto simp: rel-def dest: idinfo-crypt idinfo-mpair-fst idinfo-mpair-snd
idinfo-hash-hash)

ultimately have ∃ u v. s0 |= u ∧ u ` v ∧ v |= u1 ∧ ¬ ?P2 u ∧ ?P2 v
using B and E by (rule-tac rtrancl-start, insert
sec-prik-inj sec-prik-tok-prik, auto simp: inj-on-def image-def )

then obtain u2 v2 where H : s0 |= u2 ∧ u2 ` v2 ∧ ¬ ?P2 u2 ∧ ?P2 v2

by blast
moreover from this have Tok-PriKey n /∈ spied u2

using C by (rule-tac tok-prikey-secret, simp)
ultimately have Sec-PriKey m ∈ spied u2

proof (auto simp: rel-def dest: idinfo-crypt idinfo-mpair-fst idinfo-mpair-snd)
fix S
assume (Spy, 〈n, Hash (AgrKey (PubK S))〉) ∈ u2

moreover assume I : Sec-PriK m ∈ S
hence 〈n, Hash (PubKey S)〉 /∈ spied s0

using B and E by (insert sec-prik-inj sec-prik-tok-prik,
auto simp: inj-on-def image-def )

ultimately have ∃ u v. s0 |= u ∧ u ` v ∧ v |= u2 ∧ ¬ ?P3 S u ∧ ?P3 S v
using H by (rule-tac rtrancl-start, simp-all)

then obtain u3 v3 where s0 |= u3 ∧ u3 ` v3 ∧ v3 |= u2 ∧
¬ ?P3 S u3 ∧ ?P3 S v3

by blast
moreover from this have 〈n, PubKey S〉 ∈ spied v3

by (auto simp: rel-def dest: idinfo-crypt idinfo-mpair-fst idinfo-mpair-snd
idinfo-hash-hash)

ultimately have 〈n, PubKey S〉 ∈ spied u2

by (rule-tac subsetD [of spied v3], rule-tac spied-subset, simp)
hence Sec-PriK m /∈ S

using H by simp
thus (Spy, AgrKey (PriK (Sec-PriK m))) ∈ u2

using I by contradiction
qed
hence I : m ∈ bad-sec-prik

using H by (erule-tac contrapos-pp, rule-tac sec-prikey-secret, simp)
from B and E and G have ∃S . ?P4 S u1

by (rule-tac exI [of - {Tok-PriK n, Sec-PriK m, Rev-PriK}],
insert sec-prik-inj sec-prik-tok-prik, auto simp: inj-on-def image-def )

moreover have ¬ (∃S . ?P4 S s0)
by (insert tok-prik-rev, auto)

ultimately have ∃ u v. s0 |= u ∧ u ` v ∧ v |= u1 ∧
¬ (∃S . ?P4 S u) ∧ (∃S . ?P4 S v)
using F by (rule-tac rtrancl-start, simp)

then obtain u4 v4 S where
J : s0 |= u4 and K : u4 ` v4 and L: ¬ (∃S . ?P4 S u4) ∧ ?P4 S v4

by blast
have M : [[(Spy, 〈n, AgrKey (PubK S)〉) ∈ u4; Rev-PriK ∈ S ;
∀m. Sec-PriK m ∈ S −→ (n, m) /∈ bad-id;
∀S . Rev-PriK ∈ S −→ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4 −→

(∃m. Sec-PriK m ∈ S ∧ (n, m) ∈ bad-id)]] =⇒ False
by blast

from K and L have ∃m ′. m ′ 6= m ∧ m ′ ∈ bad-sec-prik ∧ (n, m ′) ∈ bad-id
proof (simp add: rel-def , (erule-tac disjE , (clarsimp, (erule-tac disjE ,
drule-tac sym, simp, (drule-tac idinfo-crypt [OF - J ] | drule-tac
idinfo-mpair-fst [OF - J ] | drule-tac idinfo-mpair-snd [OF - J ]), blast)+)?,
blast)+, (erule-tac disjE , erule-tac [2 ] disjE , erule-tac [3 ] disjE ; clarsimp))
fix n ′ S ′ A
assume

N : ∀S . Rev-PriK ∈ S −→ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4 −→
(∃m. Sec-PriK m ∈ S ∧ (n, m) ∈ bad-id) and

O: ∀m. Sec-PriK m ∈ S −→ (n, m) /∈ bad-id and
P: Rev-PriK ∈ S and
Q: (Spy, 〈n ′, AgrKey (PubK S ′)〉) ∈ u4 and
R: (Spy, AgrKey (PriK A)) ∈ u4

assume n = n ′ ∧ S = S ′ − {A} ∨ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4

thus ?thesis
proof (rule disjE , drule-tac [2 ] M [OF - P O N ]; clarsimp)

assume S : n = n ′ and S = S ′ − {A}
moreover from this obtain m ′ where
Sec-PriK m ′ ∈ S ′ and T : (n, m ′) ∈ bad-id
using N and P and Q by blast

ultimately have A = Sec-PriK m ′

using O by (rule-tac ccontr , simp)
hence Sec-PriKey m ′ ∈ spied u4

using R by simp
hence m ′ ∈ bad-sec-prik

by (rule contrapos-pp, rule-tac sec-prikey-secret [OF J ])
thus ∃m ′. m ′ 6= m ∧ m ′ ∈ bad-sec-prik ∧ (n ′, m ′) ∈ bad-id

using B and S and T by auto
qed

next
fix n ′ S ′ A
assume

N : ∀S . Rev-PriK ∈ S −→ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4 −→
(∃m. Sec-PriK m ∈ S ∧ (n, m) ∈ bad-id) and

O: ∀m. Sec-PriK m ∈ S −→ (n, m) /∈ bad-id and
P: Rev-PriK ∈ S and
Q: (Spy, 〈n ′, AgrKey (PubK S ′)〉) ∈ u4 and
R: (Spy, AgrKey (PriK A)) ∈ u4

assume n = n ′ ∧ S = insert A S ′ ∨ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4

thus ?thesis
proof (rule disjE , drule-tac [2 ] M [OF - P O N ]; clarsimp)

assume n = n ′ and S : S = insert A S ′

moreover have A 6= Rev-PriK
using R by (rule contrapos-pn, insert rev-prikey-secret [OF J ], simp)

ultimately obtain m ′ where Sec-PriK m ′ ∈ S ′ and T : (n, m ′) ∈ bad-id
using N and P and Q by blast

hence (n, m ′) /∈ bad-id

using O and S by simp
thus ∃m ′. m ′ 6= m ∧ m ′ ∈ bad-sec-prik ∧ (n ′, m ′) ∈ bad-id

using T by contradiction
qed

next
fix n ′ S ′

assume
N : ∀S . Rev-PriK ∈ S −→ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4 −→
(∃m. Sec-PriK m ∈ S ∧ (n, m) ∈ bad-id) and

O: ∀m. Sec-PriK m ∈ S −→ (n, m) /∈ bad-id and
P: Rev-PriK ∈ S

assume n = n ′ ∧ S = S ′ ∨ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4 and
(Spy, AgrKey (PriK (Tok-PriK n ′))) ∈ u4

thus ?thesis
by (erule-tac disjE , drule-tac [2 ] M [OF - P O N ],
insert tok-prikey-secret [OF J C ], simp-all)

next
fix n ′ X
assume

N : ∀S . Rev-PriK ∈ S −→ (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4 −→
(∃m. Sec-PriK m ∈ S ∧ (n, m) ∈ bad-id) and

O: ∀m. Sec-PriK m ∈ S −→ (n, m) /∈ bad-id and
P: Rev-PriK ∈ S and
Q: (Spy, 〈n ′, X〉) ∈ u4 ∨ (Spy, 〈n ′, Hash X〉) ∈ u4

assume n = n ′ ∧ AgrKey (PubK S) = X ∨
(Spy, 〈n, AgrKey (PubK S)〉) ∈ u4

thus ?thesis
proof (rule disjE , drule-tac [2 ] M [OF - P O N ]; clarsimp)

assume R: n = n ′ and S : X = AgrKey (PubK S)
{

assume (Spy, 〈n ′, X〉) ∈ u4

hence (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4

using R and S by simp
}
moreover {

assume (Spy, 〈n ′, Hash X〉) ∈ u4

hence ?P3 S u4

using R and S by simp
moreover have ¬ ?P3 S s0

using O by auto
ultimately have ∃ u v. s0 |= u ∧ u ` v ∧ v |= u4 ∧ ¬ ?P3 S u ∧ ?P3 S v

using J by (rule-tac rtrancl-start)
then obtain u3 v3 where s0 |= u3 ∧ u3 ` v3 ∧ v3 |= u4 ∧
¬ ?P3 S u3 ∧ ?P3 S v3

by blast
moreover from this have 〈n, AgrKey (PubK S)〉 ∈ spied v3

by (auto simp: rel-def dest: idinfo-crypt idinfo-mpair-fst idinfo-mpair-snd
idinfo-hash-hash)

ultimately have 〈n, AgrKey (PubK S)〉 ∈ spied u4

by (rule-tac subsetD [of spied v3], rule-tac spied-subset, simp)
}
ultimately have (Spy, 〈n, AgrKey (PubK S)〉) ∈ u4

using Q by blast
thus ∃m ′. m ′ 6= m ∧ m ′ ∈ bad-sec-prik ∧ (n ′, m ′) ∈ bad-id

by (drule-tac M [OF - P O N ], simp)
qed

qed
thus ?thesis

using I by simp
qed

theorem id-anonymous [rotated]:
[[m /∈ bad-sec-prik ∨ ¬ (∃m ′. m ′ 6= m ∧ m ′ ∈ bad-sec-prik ∧ (n, m ′) ∈ bad-id);

s0 |= s; (n, m) /∈ bad-id; n /∈ bad-tok-prik]] =⇒
〈n, Hash (ID n (Sec-PubKey m))〉 /∈ spied s

by (erule contrapos-pn, drule id-identified, blast+)

end

3 Possibility of anonymity compromise for token
pseudonymous identifiers

theory Possibility
imports Anonymity

begin

This section proves possibility properties tok-id-identified, sec-id-identified,
which altogether state that the spy can map a token pseudonymous identifier
to the related token if either attack option described in section 1.2 is viable.
Both properties are proven by construction, namely by creating as many
sample protocol runs such as to satisfy their conclusions if their assumptions
are fulfilled.

definition tok-id-pubk-prik :: agent-id ⇒ agent-id ⇒ state where
tok-id-pubk-prik n m ≡

insert (Spy, 〈n, ID n (Sec-PubKey m)〉) s0

definition tok-id-hash :: agent-id ⇒ agent-id ⇒ state where
tok-id-hash n m ≡

insert (Spy, 〈n, Hash (ID n (Sec-PubKey m))〉) (tok-id-pubk-prik n m)

proposition tok-id-pubk-prik-rel:
n ∈ bad-tok-prik =⇒ s0 |= tok-id-pubk-prik n m

by (subgoal-tac (s0, tok-id-pubk-prik n m) ∈ rel-id-pubk-prik,
rule r-into-rtrancl, auto simp: tok-id-pubk-prik-def rel-def image-def , blast)

proposition tok-id-pubk-prik-msg:
n ∈ bad-tok-prik =⇒

{ID n (Sec-PubKey m), Hash (ID n (Sec-PubKey m)),
〈n, ID n (Sec-PubKey m)〉} ⊆ spied (tok-id-pubk-prik n m)

by (auto simp: tok-id-pubk-prik-def )

proposition tok-id-hash-rel:
n ∈ bad-tok-prik =⇒ s0 |= tok-id-hash n m

by (rule rtrancl-into-rtrancl, erule tok-id-pubk-prik-rel [of - m],
subgoal-tac (tok-id-pubk-prik n m, tok-id-hash n m) ∈ rel-id-hash,
frule-tac [2 ] tok-id-pubk-prik-msg [of - m], auto simp: tok-id-hash-def rel-def )

theorem tok-id-identified:
n ∈ bad-tok-prik =⇒ ∃ s. s0 |= s ∧ 〈n, Hash (ID n (Sec-PubKey m))〉 ∈ spied s

by (rule exI [of - tok-id-hash n m], drule tok-id-hash-rel [of - m],
simp add: tok-id-hash-def )

definition sec-pubk-less :: agent-id ⇒ state where
sec-pubk-less n ≡

insert (Spy, PubKey {Tok-PriK n, Rev-PriK}) s0

definition sec-id-pubk-less :: agent-id ⇒ state where
sec-id-pubk-less n ≡

insert (Spy, 〈n, PubKey {Tok-PriK n, Rev-PriK}〉) (sec-pubk-less n)

definition sec-id-pubk-more :: agent-id ⇒ agent-id ⇒ state where
sec-id-pubk-more n m ≡

insert (Spy, 〈n, ID n (Sec-PubKey m)〉) (sec-id-pubk-less n)

definition sec-id-hash :: agent-id ⇒ agent-id ⇒ state where
sec-id-hash n m ≡

insert (Spy, 〈n, Hash (ID n (Sec-PubKey m))〉) (sec-id-pubk-more n m)

lemma sec-id-identified-1 :
{Tok-PriK n, Sec-PriK m, Rev-PriK} 6= {Tok-PriK n ′, Rev-PriK}

by (simp add: set-eq-iff , rule exI [of - Sec-PriK m], insert
sec-prik-rev sec-prik-tok-prik, simp add: image-def , drule spec [of - m], auto)

lemma sec-id-identified-2 :
(Spy, PubKey {Tok-PriK n, Rev-PriK}) /∈ s0

by (insert tok-prik-rev sec-id-identified-1 , simp add: image-def ,
drule spec [of - n], auto simp: set-eq-iff )

lemma sec-id-identified-3 :
{Tok-PriK n, Rev-PriK} =

{Tok-PriK n, Sec-PriK m, Rev-PriK} − {Sec-PriK m}
by (insert sec-prik-rev sec-prik-tok-prik, auto)

lemma sec-id-identified-4 :
PubK {Tok-PriK n, Sec-PriK m, Rev-PriK} =

PubK (insert (Sec-PriK m) {Tok-PriK n, Rev-PriK})
by auto

proposition sec-pubk-less-rel:
[[{m, m ′} ⊆ bad-sec-prik; (n, m) /∈ bad-id; (n, m ′) ∈ bad-id]] =⇒

s0 |= sec-pubk-less n
by (subgoal-tac (s0, sec-pubk-less n) ∈ rel-pubk-less, rule r-into-rtrancl,
simp add: rel-def , subst sec-pubk-less-def , subst sec-id-identified-3 [of - m ′],
simp, (rule exI )+, subst insert-ident, subst sec-id-identified-3 [symmetric],
insert sec-id-identified-2 , auto)

proposition sec-pubk-less-msg:
[[{m, m ′} ⊆ bad-sec-prik; (n, m) /∈ bad-id; (n, m ′) ∈ bad-id]] =⇒

{Sec-PriKey m, Sec-PriKey m ′, PubKey {Tok-PriK n, Rev-PriK},
ID n (Sec-PubKey m), Hash (ID n (Sec-PubKey m)),
〈n, ID n (Sec-PubKey m ′)〉} ⊆ spied (sec-pubk-less n) ∧

{〈n, PubKey {Tok-PriK n, Rev-PriK}〉, 〈n, ID n (Sec-PubKey m)〉,
〈n, Hash (ID n (Sec-PubKey m))〉} ∩
spied (sec-pubk-less n) = {}

by (insert sec-id-identified-2 , auto simp: sec-pubk-less-def image-def
dest: sec-prik-eq)

proposition sec-id-pubk-less-rel:
[[{m, m ′} ⊆ bad-sec-prik; (n, m) /∈ bad-id; (n, m ′) ∈ bad-id]] =⇒

s0 |= sec-id-pubk-less n
by (rule rtrancl-into-rtrancl, erule sec-pubk-less-rel, assumption+,
subgoal-tac (sec-pubk-less n, sec-id-pubk-less n) ∈ rel-id-pubk-less,
frule-tac [2 ] sec-pubk-less-msg, simp-all add: sec-id-pubk-less-def rel-def ,
(rule exI )+, subst sec-id-identified-3 [of - m ′], subst (asm) (1 2 )
sec-id-identified-3 , subst insert-ident, insert sec-prik-tok-prik, auto)

proposition sec-id-pubk-less-msg:
[[{m, m ′} ⊆ bad-sec-prik; (n, m) /∈ bad-id; (n, m ′) ∈ bad-id]] =⇒

{Sec-PriKey m, ID n (Sec-PubKey m), Hash (ID n (Sec-PubKey m)),
〈n, PubKey {Tok-PriK n, Rev-PriK}〉} ⊆
spied (sec-id-pubk-less n) ∧

{〈n, ID n (Sec-PubKey m)〉, 〈n, Hash (ID n (Sec-PubKey m))〉} ∩
spied (sec-id-pubk-less n) = {}

by (drule sec-pubk-less-msg, insert sec-id-identified-1 ,
auto simp: sec-id-pubk-less-def )

proposition sec-id-pubk-more-rel:

[[{m, m ′} ⊆ bad-sec-prik; (n, m) /∈ bad-id; (n, m ′) ∈ bad-id]] =⇒
s0 |= sec-id-pubk-more n m

by (rule rtrancl-into-rtrancl, erule sec-id-pubk-less-rel, assumption+,
subgoal-tac (sec-id-pubk-less n, sec-id-pubk-more n m) ∈ rel-id-pubk-more,
frule-tac [2 ] sec-id-pubk-less-msg, simp-all add: sec-id-pubk-more-def rel-def ,
(rule exI )+, subst sec-id-identified-4 , subst (asm) (1 3 ) sec-id-identified-4 ,
subst insert-ident, simp-all)

proposition sec-id-pubk-more-msg:
[[{m, m ′} ⊆ bad-sec-prik; (n, m) /∈ bad-id; (n, m ′) ∈ bad-id]] =⇒

{ID n (Sec-PubKey m), Hash (ID n (Sec-PubKey m)),
〈n, ID n (Sec-PubKey m)〉} ⊆ spied (sec-id-pubk-more n m) ∧

〈n, Hash (ID n (Sec-PubKey m))〉 /∈ spied (sec-id-pubk-more n m)
by (drule sec-id-pubk-less-msg, auto simp: sec-id-pubk-more-def )

proposition sec-id-hash-rel:
[[{m, m ′} ⊆ bad-sec-prik; (n, m) /∈ bad-id; (n, m ′) ∈ bad-id]] =⇒

s0 |= sec-id-hash n m
by (rule rtrancl-into-rtrancl, erule sec-id-pubk-more-rel, assumption+,
subgoal-tac (sec-id-pubk-more n m, sec-id-hash n m) ∈ rel-id-hash,
frule-tac [2 ] sec-id-pubk-more-msg, auto simp: sec-id-hash-def rel-def )

theorem sec-id-identified:
[[{m, m ′} ⊆ bad-sec-prik; (n, m ′) ∈ bad-id]] =⇒

∃ s. s0 |= s ∧ 〈n, Hash (ID n (Sec-PubKey m))〉 ∈ spied s
by (cases (n, m) ∈ bad-id, blast, rule exI [of - sec-id-hash n m],
drule sec-id-hash-rel, simp-all add: sec-id-hash-def )

end

